

NAG Library Routine Document

G01FBF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

G01FBF returns the deviate associated with the given tail probability of Student's t -distribution with real degrees of freedom, via the routine name.

2 Specification

```
FUNCTION G01FBF (TAIL, P, DF, IFAIL)
REAL (KIND=nag_wp) G01FBF
INTEGER                IFAIL
REAL (KIND=nag_wp) P, DF
CHARACTER(1)          TAIL
```

3 Description

The deviate, t_p associated with the lower tail probability, p , of the Student's t -distribution with ν degrees of freedom is defined as the solution to

$$P(T < t_p : \nu) = p = \frac{\Gamma((\nu + 1)/2)}{\sqrt{\nu\pi}\Gamma(\nu/2)} \int_{-\infty}^{t_p} \left(1 + \frac{T^2}{\nu}\right)^{-(\nu+1)/2} dT, \quad \nu \geq 1; -\infty < t_p < \infty.$$

For $\nu = 1$ or 2 the integral equation is easily solved for t_p .

For other values of $\nu < 3$ a transformation to the beta distribution is used and the result obtained from G01FEF.

For $\nu \geq 3$ an inverse asymptotic expansion of Cornish–Fisher type is used. The algorithm is described by Hill (1970).

4 References

Hastings N A J and Peacock J B (1975) *Statistical Distributions* Butterworth

Hill G W (1970) Student's t -distribution *Comm. ACM* **13(10)** 617–619

5 Arguments

1: TAIL – CHARACTER(1) *Input*

On entry: indicates which tail the supplied probability represents.

TAIL = 'U'

The upper tail probability, i.e., $P(T \geq t_p : \nu)$.

TAIL = 'L'

The lower tail probability, i.e., $P(T \leq t_p : \nu)$.

TAIL = 'S'

The two tail (significance level) probability, i.e., $P(T \geq |t_p| : \nu) + P(T \leq -|t_p| : \nu)$.

TAIL = 'C'

The two tail (confidence interval) probability, i.e., $P(T \leq |t_p| : \nu) - P(T \leq -|t_p| : \nu)$.

Constraint: TAIL = 'U', 'L', 'S' or 'C'.

- 2: P – REAL (KIND=nag_wp) *Input*
On entry: p , the probability from the required Student's t -distribution as defined by TAIL.
Constraint: $0.0 < P < 1.0$.
- 3: DF – REAL (KIND=nag_wp) *Input*
On entry: ν , the degrees of freedom of the Student's t -distribution.
Constraint: $DF \geq 1.0$.
- 4: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, because for this routine the values of the output arguments may be useful even if $IFAIL \neq 0$ on exit, the recommended value is -1 . **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**
On exit: $IFAIL = 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry $IFAIL = 0$ or -1 , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Note: G01FBF may return useful information for one or more of the following detected errors or warnings.

Errors or warnings detected by the routine:

If $IFAIL = 1, 2$ or 3 on exit, then G01FBF returns zero.

$IFAIL = 1$

On entry, $TAIL \neq 'U', 'S', 'C'$ or $'L'$.

$IFAIL = 2$

On entry, $P \leq 0.0$,
 or $P \geq 1.0$.

$IFAIL = 3$

On entry, $DF < 1.0$.

$IFAIL = 5$

Convergence in the calculation of the inverse beta value was not achieved. However, the result should be a reasonable approximation to the correct value.

$IFAIL = -99$

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

$IFAIL = -399$

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

The results should be accurate to five significant digits, for most argument values. The error behaviour for various argument values is discussed in Hill (1970).

8 Parallelism and Performance

G01FBF is not threaded in any implementation.

9 Further Comments

The value t_p may be calculated by using the transformation described in Section 3 and using G01FEF. This routine allows you to set the required accuracy.

10 Example

This example reads the probability, the tail that probability represents and the degrees of freedom for a number of Student's t -distributions and computes the corresponding deviates.

10.1 Program Text

```

Program g01fbfe

!      G01FBF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: g01fbf, nag_wp
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Real (Kind=nag_wp)          :: df, p, x
      Integer                     :: ifail
      Character (1)               :: tail
!      .. Executable Statements ..
      Write (nout,*) 'G01FBF Example Program Results'
      Write (nout,*)

!      Skip heading in data file
      Read (nin,*)

!      Display titles
      Write (nout,*) '      P      DF      TAIL      X'
      Write (nout,*)

d_lp: Do
      Read (nin,*,Iostat=ifail) p, df, tail
      If (ifail/=0) Then
         Exit d_lp
      End If

!      Calculate deviates (inverse CDF)
      ifail = -1
      x = g01fbf(tail,p,df,ifail)
      If (ifail/=0) Then
         If (ifail/=4 .And. ifail/=5) Then

```

```

        Exit d_lp
      End If
    End If

!      Display results
      Write (nout,99999) p, df, tail, x
    End Do d_lp

99999 Format (1X,2F8.3,3X,A1,3X,F8.3,A,I1)
      End Program g01fbfe

```

10.2 Program Data

```

G01FBF Example Program Data
0.0100  20.0  'S'           :P DF TAIL
0.01    7.5  'L'           :P DF TAIL
0.99   45.0  'C'           :P DF TAIL

```

10.3 Program Results

```

G01FBF Example Program Results

      P      DF      TAIL      X
0.010  20.000   S        2.845
0.010   7.500   L       -2.943
0.990  45.000   C        2.690

```
